```
(*Video 1: Optimization Problem 1 *)

(*A farmer wants to build a rectangular fenced backyard for the chickens to
 run free. The farmer only has 20m of fence available for the project. What
 is the largest area that can be fenced under those constraints? *)

In[52]:= Manipulate[
  Grid[{{Column[{"Width = ", DecimalForm[a, 4], " Height = ", DecimalForm[b, 4]}]},
    {Column[{" Area = ", DecimalForm[a b, 4],
       " Perimeter = ", DecimalForm[2 a + 2 b, 4]}]},
    {Graphics[{{White, Rectangle[{0, 0}, {10, 10}]}, EdgeForm[Thick],
       Green, Rectangle[{0, 0}, {a, b}]}]}},
   Spacings → {1, 1}, Frame → All], {{a, 1}, 0, 10}, {{b, 1}, 0, 10}]
```

Out[52]=




```
(*Constraint: Perimeter of a rectangle of width =
 a and height = b is 2a+2b == 20 *)

2 a + 2 b == 20
```

In[53]:=
```
(*Solve for b:*)
Solve[2 a + 2 b == 20, b]
```

Out[53]=
```
{{b → 10 - a}}
```

In[54]:=
```
(*Target function: Area of a rectangle of width =
 a and height = b is a*b -- would like to max*)
a * b /. {b → 10 - a}(*replace from constraint*)
```

Out[54]=
```
(10 - a) a
```

In[55]:=
```
Manipulate[
 Grid[
  {{Column[{"Width = ", DecimalForm[a, 4], " Height = ", DecimalForm[(10 - a), 4]}]},
   {Column[{" Area = ", DecimalForm[a (10 - a), 4],
      " Perimeter = ", DecimalForm[2 a + 2 (10 - a), 4]}]},
   {Graphics[{{White, Rectangle[{0, 0}, {10, 10}]}, EdgeForm[Thick],
      Green, Rectangle[{0, 0}, {a, (10 - a)}]}]}},
  Spacings → {1, 1}, Frame → All], {{a, 1}, 0, 10}]
```

Out[55]=



In[56]:=
```
(*Target function, with constraint substituted in:*)
f[a_] := (10 - a) a
```

```
(*Critical points?*)
Solve[f'[a] == 0, a]
```

Out[59]= {{a → 5}}

```
(**a=5 is the only critical point. Let's apply the second derivative test:)
```

In[60]:=
```
f''[5]
```

Out[60]= −2

```
(*f'' < 0 so a=5 is a local maximum!*)
```

In[75]:=
```
Expand[f[a]]
Plot[f[a], {a, 0, 10}]
```

Out[75]= $10\, a - a^2$

Out[76]=



In[64]:=
```
f[5]
```

Out[64]= 25

```
(*Maximum area, under the above constraints,
is achieved when a=5. Geometrically, this corresponds to the
 square being the rectangle of largest area for a fixed perimeter.*)
```

In[71]:=
```
(*This computation can be confirmed by Mathematica as follows,
without restricting domain:*)
Maximize[f[a], a]
Minimize[f[a], a]
```

Out[71]= {25, {a → 5}}

--- Minimize: The minimum is not attained at any point satisfying the given constraints.

Out[72]= {−∞, {a → −∞}}

In[73]:=
```
(*Restricting the domain:*)
Maximize[{f[a], 0 ≤ a ≤ 10}, a]
Minimize[{f[a], 0 ≤ a ≤ 10}, a]
```

Out[73]= {25, {a → 5}}

Out[74]= {0, {a → 0}}

```
(*Video 2: Optimization Problem 2 *)
```

```
(*A farmer wants to build a rectangular fenced backyard for the chickens to
 run free. The farmer only has 20m of fence available for the project but
 will build the yard adjacent to a straight river which serves as a natural
 fence. What is the largest area that can be fenced under those constraints? *)
```

In[92]:=
```
Manipulate[
 Grid[{{Column[{"Width = ", DecimalForm[a, 4], " Height = ", DecimalForm[b, 4]}]},
   {Column[{" Area = ", DecimalForm[a b, 4],
      " Perimeter = ", DecimalForm[a + 2 b, 4]}]},
   {Graphics[{{White, Rectangle[{0, 0}, {10, 10}]}, {EdgeForm[Thick], Green,
        Rectangle[{0, 0}, {a, b}]}, Blue, Rectangle[{-5, 0.2}, {15, -5}]}]}},
  Spacings → {1, 1}, Frame → All], {{a, 1}, 0, 10}, {{b, 1}, 0, 10}]
```

Out[92]=



```
(*Constraint: Total perimeter only includes 3 of the sides of the rectangle,
namely it is a+2b*)
```

In[93]:=
```
a + 2 b == 20
Solve[a + 2 b == 20, b]
```

Out[93]= $a + 2b == 20$

Out[94]= $\left\{\left\{b \to \frac{20 - a}{2}\right\}\right\}$

```
(*Target function*)
```

In[95]:= $a * b$ /. $\left\{b \to \frac{20 - a}{2}\right\}$

Out[95]= $\frac{1}{2} (20 - a)\, a$

In[96]:= g[a_] := $\frac{1}{2} (20 - a)\, a$

In[99]:=
```
Manipulate[
 Grid[
  {{Column[{"Width = ", DecimalForm[a, 4], " Height = ", DecimalForm[(20 - a)/2, 4]}]},
   {Column[{" Area = ", DecimalForm[a ((20 - a)/2), 4],
      " Perimeter = ", DecimalForm[a + 2 ((20 - a)/2), 4]}]},
   {Graphics[{{White, Rectangle[{0, 0}, {10, 10}]}, {EdgeForm[Thick], Green,
        Rectangle[{0, 0}, {a, (20 - a)/2}]}, Blue, Rectangle[{-5, 0.2}, {25, -5}]}]}},
  Spacings → {1, 1}, Frame → All], {{a, 1}, 0, 20}]
```

Out[99]=



In[102]:=
```
Expand[g[a]]
```

Out[102]= $10\,a - \frac{a^2}{2}$

In[103]:=
```
(*To find extremal points (minima/maxima) we first compute critical points:*)
Solve[g'[a] == 0, a]
```

Out[103]= {{a → 10}}

In[105]:=
```
g''[10]
```

Out[105]= -1

(*So a=10 is the maximum of the function g[a] on the interval 0≤a≤10*)

In[108]:= Plot[g[a], {a, 0, 20}]

Out[108]=



In[112]:= (*Maximum area is: *)g[10]

(*Sides achieving this are:*) $\left\{a, \frac{20-a}{2}, \frac{20-a}{2}\right\}$ /. a → 10

Out[112]= 50

Out[113]= {10, 5, 5}

(*The maximum area under these new constraints is 50m^2 and is achieved at a rectangle with one side of length a=10 and two sides of length 5*)

In[109]:= Maximize[g[a], a]

Out[109]= {50, {a → 10}}

(*Video 3: Optimization Problem 3 *)

(*Find the minimum and maximum of the function f(x) = 3x^4-4x^3+8 for x in the interval [-2,4] *)

In[128]:= Plot[3 x^4 - 4 x^3 + 8, {x, -2, 4}]

Out[128]=



```
(*Step 1: Identify target function and constraints*)
f[x_] := 3 x^4 - 4 x^3 + 8 (*target function*)
-2 ≤ x ≤ 4 (*constraints*)

(*Step 2: Find critical points of the target function,
after possibly substituting in the constraints to reduce to one variable*)

In[132]:= Solve[f'[x] == 0, x]

Out[132]= {{x → 0}, {x → 0}, {x → 1}}

In[133]:= (*Step 3: Determine the type of each critical point*)
f''[0]
f''[1]

Out[133]= 0

Out[134]= 12

(*x=1 is a local minimum*)

In[135]:= Plot[f'[x], {x, -1 / 2, 1 / 2}]

Out[135]=
```




```
(*Since tangent line has negative slope both before and after x=0,
this critical point is not a local min nor a local max*)
```

In[136]:= `Plot[f[x], {x, -1 / 2, 1 / 2}]`

Out[136]=




```
(*Step 4: Compare value of the function at local min/max inside the domain
  with values of the function at the boundary of its domain (endpoints) *)
```

In[139]:=
```
(*Left endpoint*)f[-2]
(*Right endpoint*)f[4]
```

Out[139]= 88

Out[140]= 520

In[141]:= `(*Critical points in the interior*)f[1]`

Out[141]= 7

```
(*Conclusion: the largest among these values is the global max;
the smallest is the global min.*)
```

```
(*Global minimum value is 7, achieved at the point x=1 *)
(*Global maximum value is 520, achieved at the point x=4 *)
```

In[145]:= `Plot[f[x], {x, -2, 4}]`

Out[145]=



In[146]:=
```
Maximize[{3 x^4 - 4 x^3 + 8, -2 ≤ x ≤ 4}, x]
Minimize[{3 x^4 - 4 x^3 + 8, -2 ≤ x ≤ 4}, x]
```

Out[146]= {520, {x → 4}}

Out[147]= {7, {x → 1}}